

SPSSシンタックスの基本

立教大学社会学部 村瀬 洋一

シンタックスは、SPSSにおいて使う各種の命令文（プログラム）である。これを使うと、複雑な分析やデータ加工を一度に実行できる。また実行した記録が残るため、後で同じ作業を繰り返すことも簡単にできる。SPSSを起動すると初めはデータウィンドウだけがある。

0. シンタックス・ウィンドウを開く

SPSSを起動し、画面上的「ファイル」をクリックし、新規作成→シンタックスを選ぶ




図 1. SPSS画面

この後、以下の例 1～7のように、シンタックスを書いてから実行する。シンタックスを使うと、新変数の作成などが簡単にできる。

次の図の中は、欠損値処理MISSING VALUES、COMPUTE文による回答方向を逆転した新変数作成、合計得点作成、自営業ダミー変数作成の例。これらを書き、実行したい部分を選択し▲のボタンを押して実行。シンタックスの見本ファイルを書き換えると簡単。



図 2. シンタックス・ウィンドウの例

詳しくは以下の解説をよく読むこと。

1. シンタックスによるデータ読み込み

◆SPSS形式データの場合

シンタックスによってデータを読み込む必要はない

◆テキスト形式データの場合

まずCドライブなどにデータを置く。普通ハードディスクはCドライブ。

シンタックス内の最初にある、データファイルの位置指定を、以下のように書きかえる

1.1. テキスト形式データファイルの読み込みシンタックス解説

シンタックス最初で、以下のようなDATA LIST FILE命令文を使う。データ場所の指定文として ' ' の間にデータファイル名を書く。Cドライブの DATAPR という名前のフォルダの中に SE970711.txt というファイルが入っている場合は、以下ようになる。

```
DATA LIST FILE=' C:¥DATAPR¥SE970711.txt' RECORDS=2
```

自分のUSBメモリーなどがFドライブの時はF:¥ と書く。¥で区切ってフォルダを書く。RECORDS=2は1人分が2行のデータの場合。この後に桁指定文を書きピリオドを書く。

マイドキュメント内のreiというフォルダ内にデータがある場合は以下のように書く。

```
C:¥Documents and Settings¥USERID¥My Documents¥rei¥2009abc.txt
```

USERID の部分は、Windowsにログインした時のユーザー名を書く。

フォルダやファイル名を正確に書くことが重要。全角空白や余計なピリオドを入れない。この後、シンタックスを実行すればよい。

初めデータウィンドウは空に見えるが、何らかの分析をするとデータが入る

なお、データファイルを保存する時は、ファイル名を変えずに、そのまま保存すること。

1.2. SPSS形式データファイルの読み込みシンタックス解説

SPSS形式データファイルはシンタックスを使わずにファイルを開けばよいが、以下のGET FILE命令文を使って読み込むこともできる。例えば、Hドライブに abc77.sav というファイルが入っている場合、以下ようになる（フォルダは使っていない場合の例）。

```
GET FILE ='H:¥abc77.sav' .
```

この後に桁指定文はいらない。この後にリコード文や分析命令文などを書く。

2. シンタックスの構造

普通、次の順で命令文が並ぶ。

- 1 データの場所指定文 — これによりデータファイルをSPSSに読み込む
- 2 データのケタ指定文
- 3 欠損値処理
- 4 データの加工 値の変換、新変数の作成、計算など
- 5 単純集計やクロス集計などの分析命令文

3. データの加工文

データを読み込んだ後に、カテゴリー合併や、新変数の作成、変数の計算などができる。

3.1. リコード文による値の変換

書式 RECODE 変数名 (条件式) ピリオド

例1 問3の4段階回答を2段階にする。

```
COMPUTE      N3 = Q3 .
RECODE       N3 (1, 2=1) (3, 4=2) .
MISSING VALUES N3 (9) .
```

```
/****** 上記の1行目は、新変数名(新しい質問項目)として *****/
/****** N3を設定し、その中身をQ3と同じにしている。 *****/
/****** 新変数がN3ということを理解する。N3は好きな名前で良い *****/
/****** 2行目はリコード文でのカテゴリー合併 *****/
/****** 3行目は欠損値処理。無回答9を分析から除く処理 *****/
```

これを実行しても何も起きないが、N3を使って分析するとデータにN3が追加される。

例2 AGEという細かい変数をNENDAIという5段階の変数に変換する。クロス集計の時は、あまりカテゴリーが細かいと表が読みにくいので、カテゴリー合併をすることが多い。この例では、NENDAIが新変数名だということを、まず理解すること。

```
COMPUTE      NENDAI =AGE.
RECODE       NENDAI (10 THRU 29 =2) (30 THRU 39 =3) (40 THRU 49 =4)
              (50 THRU 59 =5) (60 THRU 98 =6) .
```

例3 学歴を教育年数に変換する時は、以下のように書く。

```
COMPUTE      EDU=Q16.
RECODE       EDU (1=6) (2=9) (3=12) (4=13) (5=14) (6=16) (7, 9=99).
MISSING VALUES EDU(99).
```

3.2. IF文の例

書式 IF (条件式) 条件があった場合の命令文 ピリオド

例4 問25の職業変数をもとに、専門職ダミー変数を作成

問25が6か7の場合に、専門職ダミー変数の値を1にする

```
COMPUTE      SENMON =0.
IF (Q25=6)   SENMON =1.
IF (Q25=7)   SENMON =1.
```

例5 問1の生まれた年と月の変数をもとに、年齢の変数を作成。
生まれ月が1-4の場合に、年齢に1を足すという例。

```
COMPUTE      AGE =71-Q1NEN.
IF (Q1GETU<5) AGE=AGE+1.
```

3.3. 変数の計算 足し算やかけ算など、計算ができる。+ - * / を用いる。

例6 問4の変数の回答を逆転。N4が逆転した変数だということを理解する。
Compute N4=5-Q4.

例7 問5と問6の回答内容を足して、合計得点の新変数NEW7を作成する。
Compute NEW7=Q5+Q6.

4. おぼえておくと便利なこと

- ・シンタックスでは、大文字と小文字は区別されない
- ・シンタックスでは、半角space、改行、tabは区別されない

